

Program Term Generation Through Enumeration of Indexed datatypes (Thesis Proposal)

CAS VAN DER REST

ACM Reference Format:

Cas van der Rest. 2019. Program Term Generation Through Enumeration of Indexed datatypes (Thesis Proposal). 1, 1 (February 2019), 15 pages. <https://doi.org/10.1145/nnnnnnnn.nnnnnnnn>

Author's address: Cas van der Rest.

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than ACM must be honored. Abstracting with credit is permitted. To copy otherwise, or republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. Request permissions from permissions@acm.org.

© 2019 Association for Computing Machinery.

XXXX-XXXX/2019/2-ART \$15.00

<https://doi.org/10.1145/nnnnnnnn.nnnnnnnn>

1 INTRODUCTION

A common way of asserting a program's correctness is by defining properties that should universally hold, and asserting these properties over a range of random inputs. This technique is commonly referred to as *property based testing*, and generally consists of a two-step process. Defining properties that universally hold on all inputs, and defining *generators* that sample random values from the space of possible inputs. *QuickCheck* [5] is likely the most well known tool for performing property based tests on haskell programs.

Although coming up with a set of properties that properly captures a program's behaviour might initially seem to be the most involved part of the process, defining suitable generators for complex input data is actually quite difficult as well. Questions such as how to handle datatypes that are inhabited by an infinite number of values arise, or how to deal with constrained input data. The answers to these questions are reasonably well understood for *Algebraic datatypes (ADT's)*, but no general solution exists when more complex input data is required. In particular, little is known about enumerating and generating inhabitants of *Indexed datatypes*.

The latter may be of interest when considering property based testing in the context of languages with a more elaborate type system than Haskell's, such as *Agda* or *Idris*. Since the techniques used in existing tools such as *QuickCheck* and *SmallCheck* for the most part only apply to regular datatypes, meaning that there is no canonical way of generating inhabitants for a large class of datatypes in these languages.

Besides the obvious applications to property based testing in the context of dependently typed languages, a broader understanding of how we can generate inhabitants of indexed datatypes may prove useful in other areas as well. Since we can often capture a programming language's semantics as an indexed datatype, efficient generation of inhabitants of such a datatype may prove useful for testing compiler infrastructure.

1.1 Problem Statement

1.2 Research Questions and Contributions

What is the problem? Illustrate with an example. [2, 15]

What is/are your research questions/contributions? [5]

2 BACKGROUND

What is the existing technology and literature that I'll be studying/using in my research [8, 12, 14, 18]

2.1 Dependent Types

Dependent type theory extends a type theory with the possibility of defining types that depend on values. In addition to familiar constructs, such as the unit type (\top) and the empty type \perp , one can use so-called Π -types and Σ -types. Π -types capture the idea of dependent function types, that is, *functions* whose output type may depend on the values of its input. Given some type A and a family P of types indexed by values of type A (i.e. P has type $A \rightarrow \text{Type}$), Π -types have the following definition:

$$\Pi_{(x:A)} P(x) \quad \equiv \quad (x : A) \rightarrow P(x)$$

In a similar spirit, Σ -types are ordered *pairs* of which the type of the second value may depend on the first value of the pair.

$$\Sigma_{(x:A)} P(x) \quad \equiv \quad (x : A) \times P(x)$$

The Curry-Howard equivalence [16] extends to Π - and Σ -types as well: they can be used to model universal and existential quantification.

2.2 Agda

Agda is a programming language that implements dependent types [14]. Its syntax is broadly similar to Haskell's, though Agda's type system is vastly more expressive due to the possibility for types to depend on term level values. Agda has a dual purpose as proof assistant based on the Curry-Howard equivalence.

2.2.1 Codata and Sized Types. All definitions are required to be *total* in Agda, meaning that they should be defined and terminate in finite time on all possible inputs. The Halting problem states that it is impossible to define a general procedure that decides whether the latter condition. To ensure that only terminating definitions are accepted, Agda's termination checker uses a sound approximation. A logical consequence is that there are Agda programs that terminate, but are rejected by the termination checker. This means that we cannot work with infinite data in the same way as in the same way as in Haskell, which does not care about termination. This means that co-recursive definitions are often problematic. For example, the following definition is perfectly fine in Haskell:

```
nats :: [Int]
nats = 0 : map (+1) nats
```

meanwhile, an equivalent definition in Agda gets rejected by the Termination checker:

```
nats : List ℕ
nats = 0 :: map suc nats
```

This is no surprise, as the termination checker will reject any recursive calls where there is not at least one argument that is strictly smaller. However, in both Agda and Haskell, an expression such as `take 10 nats` evaluates to `[0, 1, ..., 9]` in finite time.

Codata. To allow these kind of manipulations on infinite structures, the Agda Standard Library makes the lazy semantics that allow these operations explicit. In the case of lists, this means that we explicitly specify that the recursive argument to the `::` constructor is a *Thunk*, which should only be evaluated when needed:

```
data Colist {a} (A : Set a) (i : Size) : Set a where
  [] : Colist A i
  _::_ : A → Thunk (Colist A) i → Colist A i
```

We can now define `nats` in Agda by wrapping the recursive call in a thunk:

```
nats : ∀ {i : Size} → Colist ℕ i
nats = 0 :: λ where .force → map suc nats'
```

Since colists are possible infinite structures, there are some functions we can define on lists, but not on colists. An example of this is a function calculating the length of a colist:

```
length : ∀ {a : Set} → Colist a ∞ → ℕ
length [] = 0
length (x :: xs) = suc (length' (xs .force))
```

Sized Types. Sized types extend the space of function definitions that are recognized by the termination checker as terminating by tracking information about the size of values in types [1]. Consider the following example of a function that increments every element in a list of naturals with its position:

```
incpos : List ℕ → List ℕ
incpos [] = []
incpos (x :: xs) = x :: incpos (map suc xs)
```

The recursive call to `incpos` gets flagged by the termination checker; we know that `map` does not alter the length of a list, but the termination checker cannot see this. For all it knows `map` equals `const [1]`, which would make `incpos` non-terminating. The size-preserving property of `map` is not reflected in its type.

We can define an alternative version of the `List` datatype indexed with `Size`, which tracks the depth of a value in its type.

```
data List (a : Set) : Size → Set where
  [] : ∀ {i} → List' a i
  _::_ : ∀ {i} → a → List' a i → List' a (↑ i)
```

here $\uparrow i$ means that the depth of a value constructed using the `::` constructor is one deeper than its recursive argument. Incidentally, the recursive depth of a list is equal to its size (or length), but this is not necessarily the case. By indexing values of `List` with their size, we can define a version of `map` which reflects in its type that the size of the input argument is preserved:

```
map : ∀ {i} {a b : Set} → (a → b) → List a i → List b i
```

using this definition of `map`, the definition of `incpos` is no longer rejected by the termination checker.

2.3 Property Based Testing

Property Based Testing aims to assert properties that universally hold for our programs by parameterizing tests over values and checking them against a collection of test values. An example of a property (in Haskell) would be:

```
reverse_preserves_length :: [a] → Bool
reverse_preserves_length xs = length xs ≡ length (reverse xs)
```

We can *check* this property by taking a collection of lists, and asserting that `reverse_preserves_length` is true on all test inputs. Libraries for property based testing often include some kind of mechanism to automatically generate collections of test values. Existing tools take different approaches towards generatino of test data: *QuickCheck* [5] randomly generates values within the test domain, while *SmallCheck* [15] and *LeanCheck* [13] exhaustively enumerate all values in the test domain up to a certain point.

2.3.1 Existing Libraries. Many libraries exist for property based testing. This section briefly discusses some of them.

QuickCheck. Published in 2000 by Claessen & Hughes [5], *QuickCheck* implements property based testing for Haskell. As mentioned before, test values are generated by sampling randomly from the domain of test values. *QuickCheck* supplies the typeclass `Arbitrary`, whose instances are those types for which random values can be generated. A property of type `a → Bool` can be

tested if `a` is an instance of `Arbitrary`. Instances for most common Haskell types are supplied by the library.

If a property fails on a testcase, QuickCheck supplies a counterexample. Consider the following faulty definition of `reverse`:

```
reverse :: Eq a => [a] -> [a]
reverse []     = []
reverse (x:xs) = nub ((reverse xs) # [x, x])
```

If we now test our function by calling `quickCheck reverse_preserves_length`, we get the following output:

```
Test.QuickCheck> quickCheck reverse_preserves_length
*** Failed! Falsifiable (after 8 tests and 2 shrinks):
[7,7]
```

We see that a counterexample was found after 8 tests *and 2 shrinks*. Due to the random nature of the tested values, the counterexamples that falsify a property are almost never minimal counterexamples. QuickCheck takes a counterexample and applies some function that produces a collection of values that are smaller than the original counterexample, and attempts to falsify the property using one of the smaller values. By repeatedly *Shrinking* a counterexample, QuickCheck is able to find much smaller counterexamples, which are in general of much more use to the programmer.

Perhaps somewhat surprising is that QuickCheck is also able to randomly generate values for function types. The general idea here is that for a function of type $a \rightarrow b$, a case expression is generated that switches over the possible constructors for `a`, and returns a random value of type `b` for every branch.

(Lazy) SmallCheck. Contrary to QuickCheck, SmallCheck [15] takes an *enumerative* approach to the generation of test data. While the approach to formulation and testing of properties is largely similar to QuickCheck's, test values are not generated at random, but rather exhaustively enumerated up to a certain *depth*. Zero-arity constructors have depth 0, while the depth of any positive arity constructor is one greater than the maximum depth of its arguments. The motivation for this is the *small scope hypothesis*: if a program is incorrect, it will almost always fail on some small input [3].

In addition to SmallCheck, there is also *Lazy SmallCheck*. In many cases, the value of a property is determined only by part of the input. Additionally, Haskell's lazy semantics allow for functions to be defined on partial inputs. The prime example of this is a property `sorted :: Ord a => [a] -> Bool` that returns `false` when presented with $1:0:\perp$. It is not necessary to evaluate \perp to determine that the input list is not ordered.

Partial values represent an entire class of values. That is, $1:0:\perp$ can be viewed as a representation of the set of lists that start with $[1, 0]$. By checking properties on partial values, it is possible to falsify a property for an entire class of values in one go, in some cases greatly reducing the amount of testcases needed.

LeanCheck. Where SmallCheck uses a value's *depth* to bound the number of test values, LeanCheck uses a value's *size* [13], where *size* is defined as the number of construction applications of positive arity.

Both SmallCheck and LeanCheck contain functionality to enumerate functions similar to QuickCheck's `Coarbitrary`.

Feat. A downside to both SmallCheck and LeanCheck is that they do not provide an efficient way to generate or sample large test values. QuickCheck has no problem with either, but QuickCheck

generators are often more tedious to write compared to their SmallCheck counterpart. Feat [9] aims to fill this gap by providing a way to efficiently enumerate algebraic types.

QuickChick.

2.3.2 Generating Test Data.

2.3.3 Automatic Generation of Specifications.

2.4 Generic Programming & Type Universes

If we desire to abstract over the structure of datatypes, we need a suitable type universe to do so. Many such universes have been developed and studied; this section discusses a few of them.

2.4.1 Regular Datatypes. The term *regular datatypes* is often used to refer to the class of datatypes that can be assembled using any combination of products, coproducts, unary constructors, constants (a position that is inhabited by a value of another type) and recursive positions.

Any value that lives in universe induced by these combinators describes a regular datatype, and is generally referred to as a *pattern functor*. We can define a datatype in agda that captures these values:

```
data Reg : Set → Set where
  U      : Reg ⊥
  K      : (a : Set) → Reg a
  _⊕_    : ∀ {a : Set} → Reg a → Reg a → Reg a
  _⊗_    : ∀ {a : Set} → Reg a → Reg a → Reg a
  I      : Reg ⊥
```

Pattern functors can be interpreted as types in such a way that inhabitants of the interpreted type correspond to inhabitants of the type that is represented by the functor.

```
[[_]] : Reg → Set → Set
[[ U      ]] r = ⊤
[[ K a    ]] r = a
[[ reg1 ⊕ reg2 ]] r = [[ reg1 ]] r ∪ [[ reg2 ]] r
[[ reg1 ⊗ reg2 ]] r = [[ reg1 ]] r × [[ reg2 ]] r
[[ I      ]] r = r
```

Notice that recursive positions are left explicit. This means that we require an appropriate fixed-point combinator:

```
data μ (f : Reg) : Set where
  'μ : [[ f ]] (μ f) → μ f
```

Example. Consider the pattern functor corresponding to the definition of *List*:

```
List' : Set → Set
List' a = μ (U ⊕ (K a ⊗ I))
```

Notice that this pattern functor denotes a choice between a unary constructor (`[[]]`), and a constructor that takes a constant of type *a* and a recursive positions as arguments (`::`). We can define conversion functions between the standard *List* type, and the interpretation of our pattern functor:

```

fromList :  $\forall \{a : \text{Set}\} \rightarrow \text{List } a \rightarrow \text{List}' a$ 
fromList [] = 'μ (inj1 tt)
fromList (x :: xs) = 'μ (inj2 (x , fromList xs))

toList :  $\forall \{a : \text{Set}\} \rightarrow \text{List}' a \rightarrow \text{List } a$ 
toList ('μ (inj1 tt)) = []
toList ('μ (inj2 (fst , snd))) = fst :: toList snd

```

Using such isomorphisms, we can automatically derive functionality for datatypes that can be captured using pattern functors. We will see an example of this in section 3.1.4, where we will derive enumeration of inhabitants for arbitrary pattern functors.

2.4.2 Ornaments. *Ornaments* [7] provide a type universe in which we can describe the structure of indexed datatypes in a very index-centric way. Indexed datatypes are described by *Signatures*, consisting of three elements:

- A function $Op : I \rightarrow \text{Set}$, that relates indices to operations/constructors
- A function $Ar : Op \ i \rightarrow \text{Set}$, that describes the arity (with respect to recursive positions) for an operation
- A typing discipline $Ty : Ar \ op \rightarrow I$, that describes indices for recursive positions.

When combined into a single structure, we say that Σ_D gives the signature of some indexed datatype $D : I \rightarrow \text{Set}$:

$$\Sigma_D(I) = \begin{cases} Op : I \rightarrow \text{Set} \\ Ar : Op \ i \rightarrow \text{Set} \\ Ty : Ar \ op \rightarrow I \end{cases}$$

Example. Let us consider the signature for the Vec type, denoted by $\Sigma_{Vec}(\mathbb{N})$. Recall the definition of the Vec datatype:

```

data Vec {a} (A : Set a) :  $\mathbb{N} \rightarrow \text{Set } a$  where
  [] : Vec A zero
  _::_ :  $\forall \{n\} (x : A) (xs : Vec A n) \rightarrow Vec A (\text{suc } n)$ 

```

It has the following relation between indices and operations (available constructors):

```

Op-vec :  $\forall \{a : \text{Set}\} \rightarrow \mathbb{N} \rightarrow \text{Set}$ 
Op-vec zero =  $\top$ 
Op-vec {a} (suc n) = a

```

If the index is *zero*, we have only the unary constructor $[]$ at our disposal, hence $\text{Op-vec } zero = \top$. If the index is *suc n*, the number of possible constructions for Vec corresponds to the set of inhabitants of its element type, hence we say that $\text{Op-vec } (suc \ n) = a$.

The $[]$ constructor has no recursive argument, so its arity is \perp . Similarly, *cons a* takes one recursive argument, so its arity is \top :

```

Ar-vec :  $\forall \{a : \text{Set}\} \rightarrow (n : \mathbb{N}) \rightarrow \text{Op-vec } \{a\} \ n \rightarrow \text{Set}$ 
Ar-vec zero tt =  $\perp$ 
Ar-vec (suc n) op =  $\top$ 

```

The definition of $::$ dictates that if the index is equal to *suc n*, the index of the recursive argument needs to be *n*. We interpret this as follows: if a vector has length $(\text{suc } n)$, its tail has length *n*. This induces the following typing discipline for Vec :

$$\begin{aligned}
\text{Ty-vec} &: \forall \{a : \text{Set}\} \rightarrow (n : \mathbb{N}) \rightarrow (\text{op} : \text{Op-vec } \{a\} n) \rightarrow \text{Ar-vec } n \text{ op} \rightarrow \mathbb{N} \\
\text{Ty-vec zero } a &() \\
\text{Ty-vec (suc } n) a \text{ tt} &= n
\end{aligned}$$

This defines the signature for Vec : $\Sigma_{\text{Vec}} \triangleq \text{Op-vec} \triangleleft^{\text{Ty-vec}} \text{Ar-vec}$.

Non-indexed datatypes can be represented as an indexed type by choosing an index type with only a single object: \top . Below is a signature definition for \mathbb{N} using \top as the index:

$$\begin{aligned}
\text{Op-nat} &: \top \rightarrow \text{Set} \\
\text{Op-nat tt} &= \top \uplus \top \\
\text{Ar-nat} &: \text{Op-nat tt} \rightarrow \text{Set} \\
\text{Ar-nat (inj}_1 x) &= \perp \\
\text{Ar-nat (inj}_2 y) &= \top \\
\text{Ty-nat} &: (\text{op} : \text{Op-nat tt}) \rightarrow \text{Ar-nat op} \rightarrow \top \\
\text{Ty-nat (inj}_1 x) &() \\
\text{Ty-nat (inj}_2 y) \text{ tt} &= \text{tt}
\end{aligned}$$

2.4.3 Functorial Species.

2.4.4 Indexed Functors.

- Libraries for property based testing (QuickCheck, (Lazy) SmallCheck, QuickChick, QuickSpec)
- Type universes (ADT's, Ornaments) [7, 10]
- Generic programming techniques. (pattern functors, indexed functors, functorial species)
- Techniques to generate complex or constrained data (Generating constrained random data with uniform distribution, Generators for inductive relations)
- Techniques to speed up generation of data (Memoization, FEAT)
- Formal specification of blockchain (bitml, (extended) UTxO ledger) [19, 20]
- Representing potentially infinite data in Agda (Colists, coinduction, sized types)

Below is a bit of Agda code:

$$\begin{aligned}
\Gamma\text{-match} &: (\tau : \text{Ty}) \rightarrow \langle\langle \omega_i (\lambda \Gamma \rightarrow \Sigma [\alpha \in \text{Id}] \Gamma [\alpha \mapsto \tau]) \rangle\rangle \\
\Gamma\text{-match } \tau \mu \emptyset &= \text{uninhabited} \\
\Gamma\text{-match } \tau \mu (\alpha \mapsto \sigma :: \Gamma) &\textbf{with } \tau \stackrel{?}{=} \sigma \\
\Gamma\text{-match } \tau \mu (\alpha \mapsto \tau :: \Gamma) \mid \text{yes refl} &= \langle (\alpha, \text{TOP}) \rangle \\
&\parallel \langle (\Sigma\text{-map POP}) (\mu \Gamma) \rangle \\
\Gamma\text{-match } \tau \mu (\alpha \mapsto \sigma :: \Gamma) \mid \text{no } \neg p &= \langle (\Sigma\text{-map POP}) (\mu \Gamma) \rangle
\end{aligned}$$

Fig. 1. Definition of Γ -match

3 PRELIMINARY RESULTS

3.1 Enumeration of Agda Types

We look at how to enumerate various datatypes in Agda, starting with simple examples such as \mathbb{N} or *Bool*, and progressively working towards more complex data. The first question we encounter is what the result of an enumeration should be. The obvious answer is that *enumerate* should return

```

data Env : Set where
  ∅ : Env
  _↦_::_ : Id → Ty → Env → Env

data _[_↦_] : Env → Id → Ty → Set where

  TOP : ∀ {Γ α τ}
    → (α ↦ τ :: Γ) [ α ↦ τ ]

  POP : ∀ {Γ α β τ σ} → Γ [ α ↦ τ ]
    → (β ↦ σ :: Γ) [ α ↦ τ ]

```

Fig. 2. Environnement definition and membership in *Agda*

$$\begin{array}{c}
 \text{TOP} \frac{}{(a \mapsto t : \Gamma)[a \mapsto t]} \qquad \text{POP} \frac{\Gamma[a \mapsto t]}{(b \mapsto s : \Gamma)[a \mapsto t]} \\
 \\
 \text{VAR} \frac{\Gamma[a \mapsto \tau]}{\Gamma \vdash a : \tau} \qquad \text{ABS} \frac{\Gamma, a \mapsto \sigma \vdash t : \tau}{\Gamma \vdash \lambda a \rightarrow t : \sigma \rightarrow \tau} \\
 \\
 \text{APP} \frac{\Gamma \vdash f : \sigma \rightarrow \tau \quad \Gamma \vdash x : \sigma}{\Gamma \vdash fx : \tau} \qquad \text{LET} \frac{\Gamma \vdash e : \sigma \quad \Gamma, a \mapsto \sigma \vdash t : \tau}{\Gamma \vdash \text{let } a := e \text{ in } t : \tau}
 \end{array}$$

Fig. 3. Semantics of the *Simply Typed Lambda Calculus*

something of type *Lista*, containing all possible values of type *a*. This is however not possible, as *List* in *Agda* can only represent a finite list, and many datatypes, such as \mathbb{N} have an infinite number of inhabitants. To solve this, we may either use the *Codata* functionality from the standard library, or index our result with some kind of metric that limits the number of solutions to a finite set. The latter approach is what is used by both *SmallCheck*[15] and *LeanCheck*[13], enumerating values up to a certain depth or size.

We admit the same approach as the *SmallCheck* library, defining an enumerator/generator to be a function of type $\mathbb{N} \rightarrow \text{List } a$, where input argument signifies the maximum depth. By working with *List*, ensuring termination becomes a lot easier, since it is by definition a finite structure. Furthermore, proving properties about generators becomes more straightforward, as we can simply prove the desired properties about the *List* type, and lift the result to our generator type.

3.1.1 Basic Combinators. We can define a few basic combinators to allow composition of generators.

Constants. Generators can yield a constant value, e.g. *true* for the *Bool* type. Unary constructors have a recursive depth of zero, so we simply return a singleton list:

```

G-pure : ∀ {a : Set} {n : ℕ} → a → G a n
G-pure x _ = [ x ]

```

Application. Many datatypes are constructed by applying a constructor to a value of another datatype. An example is the *just* constructor that takes a value of type a and yields a value of type *Maybe*. We can achieve this by lifting the familiar *map* function for lists to the generator type:

$$\begin{aligned}\mathbb{G}\text{-map} &: \forall \{a\ b : \text{Set}\} \{n : \mathbb{N}\} \rightarrow (a \rightarrow b) \rightarrow \mathbb{G}\ a\ n \rightarrow \mathbb{G}\ b\ n \\ \mathbb{G}\text{-map}\ f\ x\ n &= \text{map}\ f\ (x\ n)\end{aligned}$$

Product. When a constructor takes two or more values (e.g. $_$, $_$), enumerating all values that can be constructed using that constructor comes down to enumerating all possible combinations of its input values, and applying the constructor. Again, we can do this by defining the canonical cartesian product on lists, and lifting it to the generator type:

$$\begin{aligned}\text{list-ap} &: \forall \{\ell\} \{a\ b : \text{Set}\ \ell\} \rightarrow \text{List}\ (a \rightarrow b) \rightarrow \text{List}\ a \rightarrow \text{List}\ b \\ \text{list-ap}\ fs\ xs &= \text{concatMap}\ (\lambda\ f \rightarrow \text{map}\ f\ xs)\ fs\end{aligned}$$

$$\begin{aligned}\mathbb{G}\text{-ap} &: \forall \{a\ b : \text{Set}\} \rightarrow \mathbb{G}\ (a \rightarrow b) \rightarrow \mathbb{G}\ a \rightarrow \mathbb{G}\ b \\ \mathbb{G}\text{-ap}\ f\ x\ n &= \text{list-ap}\ (f\ n)\ (x\ n)\end{aligned}$$

Note that in addition to $\mathbb{G}\text{-ap}$, one also needs $\mathbb{G}\text{-map}$ to construct values using constructors with arity greater than one. Assuming f generates values of type a , and g generates values of type b , we can generate values of type $a \times b$ using the following snippet:

$$\begin{aligned}\text{pair} &: \forall \{a\ b : \text{Set}\} \rightarrow \mathbb{G}\ a \rightarrow \mathbb{G}\ b \rightarrow \mathbb{G}\ (a \times b) \\ \text{pair}\ f\ g &= \mathbb{G}\text{-ap}\ (\mathbb{G}\text{-map}\ _,_)\ f\ g\end{aligned}$$

Notice that $\mathbb{G}\text{-map}$, $\mathbb{G}\text{-pure}$ and $\mathbb{G}\text{-ap}$ make \mathbb{G} an instance of both *Functor* and *Applicative*, allowing us to use Agda's *idiom brackets* to define generators. This allows us to write

$$\begin{aligned}\text{pair} &: \forall \{a\ b : \text{Set}\} \{n : \mathbb{N}\} \rightarrow \mathbb{G}\ a\ n \rightarrow \mathbb{G}\ b\ n \rightarrow \mathbb{G}\ (a \times b)\ n \\ \text{pair}\ f\ g &= \llbracket f,\ g \rrbracket\end{aligned}$$

instead.

Choice. Choice between generators can be defined by first defining a *merge* function on lists

$$\begin{aligned}\text{merge} &: \forall \{\ell\} \{a : \text{Set}\ \ell\} \rightarrow \text{List}\ a \rightarrow \text{List}\ a \rightarrow \text{List}\ a \\ \text{merge}\ []\ ys &= ys \\ \text{merge}\ (x :: xs)\ ys &= x :: \text{merge}\ ys\ xs\end{aligned}$$

and lifting it to the generator type:

$$\begin{aligned}_||_ &: \forall \{a : \text{Set}\} \{n : \mathbb{N}\} \rightarrow \mathbb{G}\ a\ n \rightarrow \mathbb{G}\ a\ n \rightarrow \mathbb{G}\ a\ n \\ x\ ||\ y &= \lambda\ n \rightarrow \text{merge}\ (x\ n)\ (y\ n)\end{aligned}$$

Allowing for choice between constructors to be denoted in a very natural way:

$$\begin{aligned}\text{bool} &: \mathbb{G}\ \text{Bool} \\ \text{bool} &= \llbracket \text{true} \rrbracket \\ &\quad ||\ \llbracket \text{false} \rrbracket\end{aligned}$$

Recursion. Simply using implicit recursion is the most natural way for defining generators for recursive datatypes. However, the following definition that generates inhabitants of \mathbb{N} gets rejected by the termination checker:

$$\begin{aligned} \text{nats} &: \mathbb{G} \mathbb{N} \\ \text{nats} &= (\text{zero} \quad) \\ &\parallel (\text{suc nats} \quad) \end{aligned}$$

Though the above code does terminate, the termination checker cannot see this. Since the input depth is threaded through the applicative combinators, it is not immediately clear that the depth parameter decreases with the recursive call. We solve this by making recursive positions explicit:

$$\begin{aligned} \text{nat} &: \mathbb{G} \mathbb{N} \rightarrow \mathbb{G} \mathbb{N} \\ \text{nat } \mu &= (\text{zero} \quad) \\ &\parallel (\text{suc } \mu \quad) \end{aligned}$$

and defining an appropriate fixed-point combinator:

$$\begin{aligned} \text{fix} &: \forall \{a : \text{Set}\} \rightarrow (\mathbb{G} a \rightarrow \mathbb{G} a) \rightarrow \mathbb{G} a \\ \text{fix } f \ 0 &= [] \\ \text{fix } f \ (\text{suc } n) &= f (\text{fix } f) \ n \end{aligned}$$

This definition of *fix* gets rejected by the termination checker as well. We will see later how we can fix this. However, it should be apparent that it is terminating under the assumption that *f* is well-behaved, i.e. it applies the *n* supplied by *fix* to its recursive positions.

3.1.2 Indexed Types. Indexed types can be generated as well. Indexed generators can simply be defined as a Π -type, where the generated type depends on some input index:

$$\begin{aligned} \mathbb{G}_i &: \forall \{i : \text{Set}\} \rightarrow (i \rightarrow \text{Set}) \rightarrow \text{Set} \\ \mathbb{G}_i \{i = i\} \ a &= (x : i) \rightarrow \mathbb{G} (a \ x) \end{aligned}$$

The previously defined combinators can then be easily lifted to work with indexed types:

$$\begin{aligned} _||_{i_} &: \forall \{i : \text{Set}\} \{a : i \rightarrow \text{Set}\} \rightarrow \mathbb{G}_i a \rightarrow \mathbb{G}_i a \rightarrow \mathbb{G}_i a \\ (f ||_i g) \ i &= f \ i \ || \ g \ i \end{aligned}$$

Throughout the code, a subscript *i* is used to indicate that we deal with indexed types.

3.1.3 Guaranteeing Termination. We can prove termination for our fixed-point combinator if we somehow enforce that its input function is well behaved. Consider the following example of a generator that does not terminate under our fixed-point combinator:

$$\begin{aligned} \text{bad} &: \mathbb{G} \mathbb{N} \rightarrow \mathbb{G} \mathbb{N} \\ \text{bad } \mu _ &= \text{map suc } (\mu \ 1) \end{aligned}$$

Clearly, the base case of *fix* is never reached. We can solve this by indexing generators with a natural number, and requiring generators to be called with their index, yielding the following alternative definition for \mathbb{G} :

$$\begin{aligned} \mathbb{G} &: \text{Set} \rightarrow \mathbb{N} \rightarrow \text{Set} \\ \mathbb{G} \ a \ m &= (p : \Sigma [n \in \mathbb{N}] \ n \equiv m) \rightarrow \text{List } a \end{aligned}$$

We then use the following type for recursive generators:

$$\begin{aligned} \langle\langle_ \rangle\rangle &: (\mathbb{N} \rightarrow \text{Set}) \rightarrow \text{Set} \\ \langle\langle a \rangle\rangle &= \forall \{n : \mathbb{N}\} \rightarrow a \ n \rightarrow a \ n \end{aligned}$$

Meaning that the resulting generator can only apply *its own input number* to recursive positions. If we now decrease the index explicitly in the fixed-point combinator, the termination checker is able to see that *fix* always terminates.

```
fix : ∀ {a : Set} → (n : ℕ) → ⟨⟨ ℚ a ⟩⟩ → ℚ a n
fix zero    f (.0 , refl)    = []
fix (suc n) f (.suc n , refl) = f {n} (fix n f) (n , refl)
```

Let us reconsider the previous counterexample:

```
bad : ⟨⟨ ℚ ℕ ⟩⟩
bad μ n = map suc (μ (1 , {!!}))
```

It is impossible to complete this definition when applying any other value than *n* to the recursive position.

3.1.4 Deriving Enumeration for Regular Types. One may have noticed that the way in which generators are defined is structurally *very* similar to how one would define the corresponding datatypes in Haskell. This similarity is intentional, and serves to illustrate that the definition of many generators is completely mechanical with respect to the structure of the underlying datatype.

If we consider the universe of regular datatypes described in section 2.4.1, we see that there is a clear correspondence between our generator combinators, and the constructors of the *Reg* datatype. We can utilize this correspondence to automatically derive generators for datatypes, given an isomorphism with the fixed-point of some pattern functor.

Generating pattern functors. Recall that by fixing the interpretation of some value *f* of type *Reg*, we get a type whose inhabitants correspond to the inhabitants of the type that is represented by *f*. If we thus construct a generator that produces all inhabitants of this type, we have a generator that is isomorphic to a complete generator for the type represented by *f*. Doing this generically amounts to constructing a function of the following type:

```
deriveGen : (f : Reg) → ⟨⟨ ℚ (μ f) ⟩⟩
deriveGen = {!!}
```

Intuitively, this definition is easily completed by pattern matching on *f*, and returning the appropriate combinator. However, due to the intertwined usage of two fixed-point combinators to deal with recursion, there are quite a few subtleties that need to be taken into account.

We simplify the definition slightly by expanding the generator type: μ has one constructor, with one argument, so we replace μf by its constructor's argument: $\llbracket f \rrbracket (\mu f)$.

Let us now consider the branch of *deriveGen* that deals with coproducts. We would like to simply write the following:

```
deriveGen (f1 ⊕ f2) μ = (⟨ in1 (deriveGen f1 μ) ⟩) || (⟨ in2 (deriveGen f2 μ) ⟩)
```

This definition is incorrect, however. The recursive call *deriveGen* *f*₁ yields a generator of type $\langle\langle \mathbb{Q} (\llbracket f_1 \rrbracket (\mu f_1)) \rangle\rangle$, meaning that two things go wrong: The recursive argument μ we apply to the recursive call has the wrong type, and recursive positions in *f*₁ refer to values of type μf_1 instead of $\mu (f_1 \oplus f_2)$. A similar problem occurs when attempting to define a suitable definition for products.

We solve this issue by *remembering* the top-level pattern functor for which we are deriving a generator when entering recursive calls to *deriveGen*. This can be done by having the recursive argument be a generator for the interpretation of this top-level pattern functor:

```
deriveGen : ∀ {n : ℕ} → (f g : Reg) → ℚ (⟨ g ⟩) (μ g) n → ℚ (⟨ f ⟩) (μ g) n
```

By using the type signature defined above instead, the previously shown definition for the coproduct branch is accepted.

In most cases, the initial call to *deriveGen* will have the same value for f and g . Observe that $\forall f \in \text{Reg} . \text{deriveGen } f \ f : \mathbb{G} (\llbracket f \rrbracket (\mu f)) \ n \rightarrow \mathbb{G} (\llbracket f \rrbracket (\mu f)) \ n$, thus we can use *fix* to obtain a generator that generates values of type $\llbracket f \rrbracket (\mu f)$.

Deriving generators from isomorphism. We use the following record to witness an isomorphism between type a and b :

```
record _≅_ (a b : Set) : Set where
  field
    from : a → b
    to    : b → a
    iso1 : ∀ {x : a} → to (from x) ≡ x
    iso2 : ∀ {y : b} → from (to y) ≡ y
```

The functions *from* and *to* allow for conversion between a and b , while *iso₁* and *iso₂* assert that these conversion functions do indeed form a bijection between values of type a and type b . Given an isomorphism $a \cong b$, a generator $\mathbb{G} \ a \ n$ can easily be converted to a generator $\mathbb{G} \ b \ n$ by using $(_ \equiv _ . \text{to } \text{gen})$.

We can say that some type a is Regular if there exists some value f of type *Reg* such that a is isomorphic to μf . We capture this notion using the following record:

```
record Regular (a : Set) : Set where
  field
    W : Σ[ f ∈ Reg ] (a ≅ μ f)
```

Given a value of type *Regular* a , we can now derive a generator for a by deriving a generator for f , and traveling through the isomorphism by applying the aforementioned conversion.

3.2 Proving Correctness of Generators

Since generators are essentially an embellishment of the *List* monad, we can reasonably expect them to behave according to our expectations. However, it would be better to prove that generators behave as intended. Before we can start reasoning about generators, we need to formulate our properties of interest:

Productivity. We say that a generator g produces some value x if there exists some $n \in \mathbb{N}$ such that x is an element of gn . We denote this by $g \rightsquigarrow x$. Below is the Agda formulation for this property:

```
_↪_ : ∀ {a : Set} → (∀ {n : ℕ} → G a n) → a → Set
f ↪ x = ∃[ n ] (x ∈ f (n , refl))
```

Completeness. A generator $g : \mathbb{G} \ a \ n$ is complete when for all $x : a$, $g \rightsquigarrow x$. Informally, this means that a complete generator will eventually produce any inhabitant of the type it generates, provided it is given a large enough depth bound. We can formulate this in Agda as follows:

```
Complete : ∀ {a : Set} → (∀ {n : ℕ} → G a n) → Set
Complete {a} f = ∀ {x : a} → f ↪ x
```

Equivalence. Informally, two generators of type $\mathbb{G} a n$ can be considered equivalent if they produce the same elements. We formulate this as a bi-implication between productivity proofs, i.e. for all $x : a$, $g_1 \rightsquigarrow x$ if and only if $g_2 \rightsquigarrow x$. In Agda:

$$\begin{aligned} _ \rightsquigarrow _ &: \forall \{a\} (g_1 g_2 : \forall \{n\} \rightarrow \mathbb{G} a n) \rightarrow \text{Set} \\ g_1 \rightsquigarrow g_2 &= (\forall \{x\} \rightarrow g_1 \rightsquigarrow x \rightarrow g_2 \rightsquigarrow x) \times (\forall \{x\} \rightarrow g_2 \rightsquigarrow x \rightarrow g_1 \rightsquigarrow x) \end{aligned}$$

Notice that equivalence follows trivially from completeness, i.e. if two generators produce the same type, and they are both complete, then they are equivalent:

$$\begin{aligned} \text{Complete} \rightarrow \text{eq} &: \forall \{a\} \{g_1 g_2 : \forall \{n\} \rightarrow \mathbb{G} a n\} \\ &\rightarrow \text{Complete } g_1 \rightarrow \text{Complete } g_2 \\ &\rightarrow g_1 \rightsquigarrow g_2 \\ \text{Complete} \rightarrow \text{eq } p_1 p_2 &= (\lambda _ \rightarrow p_2), (\lambda _ \rightarrow p_1) \end{aligned}$$

3.2.1 Combinator Correctness. A natural starting point is to prove that properties are preserved by combinators. This section is by no means intended to exhaustively enumerate all possible combinations of combinators and properties and prove them correct, but rather serves to illustrate the general structure which can be used to construct such proofs.

We take productivity of choice as an example, hence our goal is to show that if, for some generator $g_1 : \mathbb{G} a n$ and $x : a$, $g_1 \rightsquigarrow x$, then for all generators g_2 we have that $(g_1 \parallel g_2) \rightsquigarrow x$. Since the \parallel -combinator is defined in terms of *merge*, we first prove a similar property over the *merge* function.

$$\begin{aligned} \text{merge-complete-left} &: \forall \{\ell\} \{a : \text{Set } \ell\} \{xs\ ys : \text{List } a\} \{x : a\} \\ &\rightarrow x \in xs \\ &\rightarrow x \in \text{merge } xs\ ys \\ \text{merge-complete-left (here)} &= \text{here} \\ \text{merge-complete-left } \{xs = _ :: xs\} \text{ (there } p) &= \\ \text{merge-cong } \{xs = xs\} \text{ (merge-complete-left } p) & \end{aligned}$$

merge-cong is a lemma stating that if $y \in \text{merge } xs\ ys$, then $y \in \text{merge } (x :: xs)\ ys$; its definition is omitted for conciseness. Armed with the above lemma that asserts left-completeness of the *merge* function, we can set out to prove left-completeness for the \parallel -combinator. The key insight here is that the depth bound at which x occurs does not change, thus we can simply reuse it, and lift the above lemma to the generator type:

$$\begin{aligned} \parallel\text{-complete-left} &: \forall \{a : \text{Set}\} \{x : a\} \{f g : \forall \{n : \mathbb{N}\} \rightarrow \mathbb{G} a n\} \\ &\rightarrow f \rightsquigarrow x \\ &\rightarrow (f \parallel g) \rightsquigarrow x \\ \parallel\text{-complete-left } (n, p) &= n, \text{merge-complete-left } p \end{aligned}$$

We can construct a similar proof for products by first proving similar properties about lists, and lifting them to the generator type. Proofs about the productivity of combinators can, in a similar fashion, consequently be lifted to reason about completeness. This allows us to show that, for example, if the two operands of a choice are both complete, then the resulting generator is complete as well.

3.2.2 Correctness of Derived Generators.

3.3 Generalization to Indexed Types

What examples can you handle already? [11]

What prototype have I built? [6, 9]

How can I generalize these results? What problems have I identified or do I expect? [17]

4 TIMETABLE AND PLANNING

What will I do with the remainder of my thesis? [4]

Give an approximate estimation/timetable for what you will do and when you will be done.

REFERENCES

- [1] ABEL, A. Miniagda: Integrating sized and dependent types. *arXiv preprint arXiv:1012.4896* (2010).
- [2] ALTENKIRCH, T., AND McBRIDE, C. Generic programming within dependently typed programming. In *Generic Programming*. Springer, 2003, pp. 1–20.
- [3] ANDONI, A., DANILIUC, D., KHURSHID, S., AND MARINOV, D. Evaluating the “small scope hypothesis”. In *In Popl* (2003), vol. 2, Citeseer.
- [4] CLAESSEN, K., DUREGÅRD, J., AND PÅLKA, M. H. Generating constrained random data with uniform distribution. *Journal of functional programming* 25 (2015).
- [5] CLAESSEN, K., AND HUGHES, J. Quickcheck: a lightweight tool for random testing of haskell programs. *Acm sigplan notices* 46, 4 (2011), 53–64.
- [6] CLAESSEN, K., SMALLBONE, N., AND HUGHES, J. Quickspec: Guessing formal specifications using testing. In *International Conference on Tests and Proofs* (2010), Springer, pp. 6–21.
- [7] DAGAND, P.-É. The essence of ornaments. *Journal of Functional Programming* 27 (2017).
- [8] DÉNÈS, M., HRITCU, C., LAMPROPOULOS, L., PARASKEVOPOULOU, Z., AND PIERCE, B. C. Quickchick: Property-based testing for coq. In *The Coq Workshop* (2014).
- [9] DUREGÅRD, J., JANSSON, P., AND WANG, M. Feat: functional enumeration of algebraic types. *ACM SIGPLAN Notices* 47, 12 (2013), 61–72.
- [10] KO, H.-S., AND GIBBONS, J. Programming with ornaments. *Journal of Functional Programming* 27 (2016).
- [11] LAMPROPOULOS, L., PARASKEVOPOULOU, Z., AND PIERCE, B. C. Generating good generators for inductive relations. *Proceedings of the ACM on Programming Languages* 2, POPL (2017), 45.
- [12] LÖH, A., AND MAGALHAES, J. P. Generic programming with indexed functors. In *Proceedings of the seventh ACM SIGPLAN workshop on Generic programming* (2011), ACM, pp. 1–12.
- [13] MATELA BRAQUEHAIS, R. *Tools for Discovery, Refinement and Generalization of Functional Properties by Enumerative Testing*. PhD thesis, University of York, 2017.
- [14] NORELL, U. Dependently typed programming in agda. In *International School on Advanced Functional Programming* (2008), Springer, pp. 230–266.
- [15] RUNCIMAN, C., NAYLOR, M., AND LINDBLAD, F. Smallcheck and lazy smallcheck: automatic exhaustive testing for small values. In *Acm sigplan notices* (2008), vol. 44, ACM, pp. 37–48.
- [16] WADLER, P. Propositions as types. *Communications of the ACM* 58, 12 (2015), 75–84.
- [17] YAKUSHEV, A. R., HOLDERMANS, S., LÖH, A., AND JEURING, J. Generic programming with fixed points for mutually recursive datatypes. In *ACM Sigplan Notices* (2009), vol. 44, ACM, pp. 233–244.
- [18] YORGEY, B. A. Species and functors and types, oh my! In *ACM Sigplan Notices* (2010), vol. 45, ACM, pp. 147–158.
- [19] ZAHNENTFERNER, J. An abstract model of utxo-based cryptocurrencies with scripts. *IACR Cryptology ePrint Archive* 2018 (2018), 469.
- [20] ZAHNENTFERNER, J., AND HK, I. O. Chimeric ledgers: Translating and unifying utxo-based and account-based cryptocurrencies. Tech. rep., Cryptology ePrint Archive, Report 2018/262, 2018. <https://eprint.iacr.org> ..., 2018.